Maple support for the study of alternative methods in Section 8.

Bartlett's version 1, based on Renshaw (3.5.21), with s = 1 .

```
> restart;
```

The first two cumulant equations used by Renshaw are found by putting the derivatives in (3.4.13), (3.4.14) equal to zero. The third one is (3.5.21).
I denote them by A, B, Ca:

```
> A:=(a-b*kappa1)*kappa1 - b*kappa2;
  B:=(c-d*kappa1)*kappa1 + (2*a-d-4*b*kappa1)*kappa2 - 2*b*kappa3;
  Ca:=-b*kappa4 + (a-4*b*kappa1)*kappa3 + (3*a*kappa1-3*b*kappa2-6*
  b*kappa1^2)*kappa2
     + (a*kappa1^2-b*kappa1^3+c*kappa1-d*kappa1^2)*kappa1;
```

$$A := (a - b\,\kappa1)\,\kappa1 - b\,\kappa2$$

$$B := (c - d\,\kappa1)\,\kappa1 + (2\,a - d - 4\,b\,\kappa1)\,\kappa2 - 2\,b\,\kappa3$$

$$Ca := -b\,\kappa4 + (a - 4\,b\,\kappa1)\,\kappa3 + \left(3\,a\,\kappa1 - 3\,b\,\kappa2 - 6\,b\,\kappa1^2\right)\kappa2 + \left(a\,\kappa1^2 - b\,\kappa1^3 + c\,\kappa1 - d\,\kappa1^2\right)\kappa1 \quad (1)$$

```
> A:=map(simplify,collect(A,[a,b]));
  B:=map(simplify,collect(B,[a,b,c,d]));
  Ca:=map(simplify,collect(Ca,[a,b,c,d]));
```

$$A := a\,\kappa1 - \left(\kappa1^2 + \kappa2\right) b$$

$$B := 2\,\kappa2\,a - 2\,(2\,\kappa1\,\kappa2 + \kappa3)\,b + c\,\kappa1 - \left(\kappa1^2 + \kappa2\right) d$$

$$Ca := \left(\kappa3 + 3\,\kappa1\,\kappa2 + \kappa1^3\right) a - \left(\kappa4 + 4\,\kappa1\,\kappa3 + 3\,\kappa2^2 + 6\,\kappa2\,\kappa1^2 + \kappa1^4\right) b + \kappa1^2\,c - d\,\kappa1^3 \quad (2)$$

Reparametrize with s=1 and mu=1:

```
> a:=R0-1;
  b:=(R0+alpha)/N;
  c:=R0+1;
  d:=(R0-alpha)/N;
```

$$a := R0 - 1$$

$$b := \frac{R0 + \alpha}{N}$$

$$c := R0 + 1$$

$$d := \frac{R0 - \alpha}{N} \quad (3)$$

```
> A;
  B;
```

```
Ca;
```

$$(R0-1)\,\kappa 1-\frac{\left(\kappa 1^{2}+\kappa 2\right)\,(R0+\alpha)}{N}$$

$$2\,\kappa 2\,(R0-1)-\frac{2\,(2\,\kappa 1\,\kappa 2+\kappa 3)\,(R0+\alpha)}{N}+(R0+1)\,\kappa 1-\frac{\left(\kappa 1^{2}+\kappa 2\right)\,(R0-\alpha)}{N}$$

$$\left(\kappa 3+3\,\kappa 1\,\kappa 2+\kappa 1^{3}\right)\,(R0-1)-\frac{\left(\kappa 4+4\,\kappa 1\,\kappa 3+3\,\kappa 2^{2}+6\,\kappa 2\,\kappa 1^{2}+\kappa 1^{4}\right)\,(R0+\alpha)}{N}$$

$$+\kappa 1^{2}\,(R0+1)-\frac{(R0-\alpha)\,\kappa 1^{3}}{N} \quad \textbf{(4)}$$

Ansats:

```
> kappa1:=x1*N+x2+x3/N;
  kappa2:=y1*N+y2;
  kappa3:=z1*N;
  kappa4:=u1*N;
```

$$\kappa 1:=x1\,N+x2+\frac{x3}{N}$$

$$\kappa 2:=y1\,N+y2$$

$$\kappa 3:=z1\,N$$

$$\kappa 4:=u1\,N \quad \textbf{(5)}$$

```
> A;
  B;
  Ca;
```

$$(R0-1)\left(x1\,N+x2+\frac{x3}{N}\right)-\frac{\left(\left(x1\,N+x2+\frac{x3}{N}\right)^{2}+y1\,N+y2\right)\,(R0+\alpha)}{N}$$

$$2\,(y1\,N+y2)\,(R0-1)-\frac{2\left(2\left(x1\,N+x2+\frac{x3}{N}\right)(y1\,N+y2)+z1\,N\right)(R0+\alpha)}{N}$$

$$+(R0+1)\left(x1\,N+x2+\frac{x3}{N}\right)-\frac{\left(\left(x1\,N+x2+\frac{x3}{N}\right)^{2}+y1\,N+y2\right)\,(R0-\alpha)}{N}$$

$$\left(z1\,N+3\left(x1\,N+x2+\frac{x3}{N}\right)(y1\,N+y2)+\left(x1\,N+x2+\frac{x3}{N}\right)^{3}\right)(R0-1) \quad \textbf{(6)}$$

$$-\frac{1}{N}\Bigg(\Bigg(u1\,N+4\left(x1\,N+x2+\frac{x3}{N}\right)z1\,N+3\,(y1\,N+y2)^{2}+6\,(y1\,N$$

$$+y2)\left(x1\,N+x2+\frac{x3}{N}\right)^{2}+\left(x1\,N+x2+\frac{x3}{N}\right)^{4}\Bigg)(R0+\alpha)\Bigg)+\Bigg(x1\,N+x2$$

$$+\frac{x3}{N}\Big)^2 (R0+1) - \frac{(R0-\alpha)\left(x1\,N + x2 + \frac{x3}{N}\right)^3}{N}$$

```
> A1:=coeff(A,N,1);
  A2:=coeff(A,N,0);
  A3:=coeff(A,N,-1);
```

$$A1 := (R0-1)\,x1 - x1^2\,(R0+\alpha)$$

$$A2 := (R0-1)\,x2 - (2\,x2\,x1 + y1)\,(R0+\alpha)$$

$$A3 := (R0-1)\,x3 - \left(2\,x3\,x1 + x2^2 + y2\right)(R0+\alpha) \quad \textbf{(7)}$$

```
> B1:=coeff(B,N,1);
  B2:=coeff(B,N,0);
```

$$B1 := 2\,y1\,(R0-1) - 4\,x1\,y1\,(R0+\alpha) + (R0+1)\,x1 - x1^2\,(R0-\alpha)$$

$$B2 := 2\,y2\,(R0-1) - 2\,(2\,x2\,y1 + 2\,x1\,y2 + z1)\,(R0+\alpha) + (R0+1)\,x2 - (2\,x2\,x1 + y1)\,(R0-\alpha) \quad \textbf{(8)}$$

```
> Ca3:=coeff(Ca,N,3);
  Ca2:=coeff(Ca,N,2);
  Ca1:=coeff(Ca,N,1);
```

$$Ca3 := x1^3\,(R0-1) - x1^4\,(R0+\alpha)$$

$$Ca2 := \left(3\,x1\,y1 + 3\,x2\,x1^2\right)(R0-1) - \left(6\,y1\,x1^2 + 4\,x2\,x1^3\right)(R0+\alpha) + x1^2\,(R0+1) - (R0-\alpha)\,x1^3$$

$$Ca1 := \left(z1 + 3\,x2\,y1 + 3\,x1\,y2 + x3\,x1^2 + 2\,x2^2\,x1 + x1\left(2\,x3\,x1 + x2^2\right)\right)(R0-1) - \left(4\,x1\,z1 + 3\,y1^2 + 6\,y2\,x1^2 + 12\,y1\,x2\,x1 + 2\left(2\,x3\,x1 + x2^2\right)x1^2 + 4\,x2^2\,x1^2\right)(R0+\alpha) + 2\,x2\,x1\,(R0+1) - 3\,(R0-\alpha)\,x2\,x1^2 \quad \textbf{(9)}$$

Solve A1 for x1, and exclude the spurious solution 0. Then solve B1 for y1, and then solve A2 for x2:

```
> x1a:=solve(A1,x1);
  x1:=x1a[2];
```

$$x1a := 0,\ \frac{R0-1}{R0+\alpha}$$

$$x1 := \frac{R0-1}{R0+\alpha} \quad \textbf{(10)}$$

```
> y1:=solve(B1,y1);
```

$$y1 := \frac{R0\,(\alpha+1)}{(R0+\alpha)^2} \quad \textbf{(11)}$$

```
> x2:=solve(A2,x2);
```

$$x2 := -\frac{R0\,(\alpha+1)}{(R0+\alpha)\,(R0-1)} \quad \textbf{(12)}$$

I insert the expressions that I have found for x1, x2, y1 into A3, B2, Ca3, Ca2, Ca1.
I show then that Ca3 and Ca2 are identically equal to zero.
A3, B2, and Ca1 are then simplified.
After this, I use A3, B2, Ca1 to solve for x3, y2, z1. I do this by solving A3 for x3 as fcn of y2, then solving B2 for z1 as fcn of y2, finally solving Ca1 for y2.

```
> Ca3;
```

$$0 \quad (13)$$

```
> Ca2;
  simplify(Ca2);
```

$$-\frac{2\,R0\,(\alpha+1)\,(R0-1)^2}{(R0+\alpha)^3}+\frac{(R0-1)^2\,(R0+1)}{(R0+\alpha)^2}-\frac{(R0-\alpha)\,(R0-1)^3}{(R0+\alpha)^3}$$

$$0 \quad (14)$$

```
> A3;
```

$$(R0-1)\,x3-\left(\frac{2\,x3\,(R0-1)}{R0+\alpha}+\frac{R0^2\,(\alpha+1)^2}{(R0+\alpha)^2\,(R0-1)^2}+y2\right)(R0+\alpha) \quad (15)$$

I simplify:

```
> A3a:=collect(A3,[x3,y2]);
```

$$A3a:=(-R0+1)\,x3+(-R0-\alpha)\,y2-\frac{R0^2\,(\alpha+1)^2}{(R0+\alpha)\,(R0-1)^2} \quad (16)$$

Check that A3=A3a:

```
> simplify(A3-A3a);
```

$$0 \quad (17)$$

```
> B2;
```

$$2\,y2\,(R0-1)-2\left(-\frac{2\,R0^2\,(\alpha+1)^2}{(R0+\alpha)^3\,(R0-1)}+\frac{2\,(R0-1)\,y2}{R0+\alpha}+z1\right)(R0+\alpha)$$
$$-\frac{(R0+1)\,R0\,(\alpha+1)}{(R0+\alpha)\,(R0-1)}+\frac{R0\,(\alpha+1)\,(R0-\alpha)}{(R0+\alpha)^2} \quad (18)$$

Simplify:

```
> B2a:=collect(B2,[y2,z1]);
```

$$B2a:=(-2\,R0+2)\,y2+(-2\,R0-2\,\alpha)\,z1+\frac{R0\,(\alpha+1)\,(R0-\alpha)}{(R0+\alpha)^2}$$
$$+\frac{4\,R0^2\,(\alpha+1)^2}{(R0+\alpha)^2\,(R0-1)}-\frac{(R0+1)\,R0\,(\alpha+1)}{(R0+\alpha)\,(R0-1)} \quad (19)$$

```
> B2b:=op(1,B2a) + op(2,B2a) + simplify(op(3,B2a)+op(4,B2a)+op(5,
  B2a));
```

$$B2b:=(-2\,R0+2)\,y2+(-2\,R0-2\,\alpha)\,z1+\frac{2\,R0^2\,(\alpha+1)^2}{(R0+\alpha)^2\,(R0-1)} \quad (20)$$

Check that B2=B2b:

```
> simplify(B2-B2b);
```

$$0 \quad (21)$$

```
> Ca1;
```

$$\left(z1-\frac{R0^2(\alpha+1)^2}{(R0+\alpha)^3(R0-1)}+\frac{3(R0-1)y2}{R0+\alpha}+\frac{x3(R0-1)^2}{(R0+\alpha)^2}+\frac{(R0-1)\left(\frac{2x3(R0-1)}{R0+\alpha}+\frac{R0^2(\alpha+1)^2}{(R0+\alpha)^2(R0-1)^2}\right)}{R0+\alpha}\right)(R0-1)$$
$$-\left(\frac{4(R0-1)z1}{R0+\alpha}-\frac{5R0^2(\alpha+1)^2}{(R0+\alpha)^4}+\frac{6y2(R0-1)^2}{(R0+\alpha)^2}+\frac{2\left(\frac{2x3(R0-1)}{R0+\alpha}+\frac{R0^2(\alpha+1)^2}{(R0+\alpha)^2(R0-1)^2}\right)(R0-1)^2}{(R0+\alpha)^2}\right)(R0+\alpha)$$
$$-\frac{2R0(\alpha+1)(R0+1)}{(R0+\alpha)^2}+\frac{3(R0-\alpha)R0(\alpha+1)(R0-1)}{(R0+\alpha)^3} \quad (22)$$

```
> Ca1a:=collect(Ca1,[x3,y2,z1]);
```

$$Ca1a:=-\frac{(R0-1)^3x3}{(R0+\alpha)^2}-\frac{3(R0-1)^2y2}{R0+\alpha}+(-3R0+3)z1-\frac{2R0(\alpha+1)(R0+1)}{(R0+\alpha)^2}+\frac{3(R0-\alpha)R0(\alpha+1)(R0-1)}{(R0+\alpha)^3}+\frac{3R0^2(\alpha+1)^2}{(R0+\alpha)^3} \quad (23)$$

```
> Ca1b:=op(1,Ca1a)+op(2,Ca1a)+op(3,Ca1a)+simplify(op(4,Ca1a)+op(5,
  Ca1a)+op(6,Ca1a));
```

$$Ca1b:=-\frac{(R0-1)^3x3}{(R0+\alpha)^2}-\frac{3(R0-1)^2y2}{R0+\alpha}+(-3R0+3)z1+\frac{R0(\alpha+1)(R0^2-2R0\alpha-2R0+\alpha)}{(R0+\alpha)^3} \quad (24)$$

Check that Ca1= Ca1b:

```
> simplify(Ca1-Ca1b);
```

$$0 \quad (25)$$

Solve A3a for x3 as function of y2:

```
> x3a:=solve(A3a,x3);
```

(26)

$$x3a := -\frac{1}{(R0-1)^3\,(R0+\alpha)}\left(y2\,R0^4+2\,y2\,R0^3\,\alpha+y2\,R0^2\,\alpha^2-2\,y2\,R0^3-4\,y2\,R0^2\,\alpha-2\,y2\,R0\,\alpha^2+y2\,R0^2+2\,y2\,R0\,\alpha+y2\,\alpha^2+R0^2\,\alpha^2+2\,R0^2\,\alpha+R0^2\right) \quad (26)$$

Simplify:

```
> x3:=-(alpha+1)^2*R0^2/(R0+alpha)/(R0-1)^3 - y2*(R0+alpha)/(R0-1);
```

$$x3 := -\frac{(\alpha+1)^2\,R0^2}{(R0-1)^3\,(R0+\alpha)}-\frac{y2\,(R0+\alpha)}{R0-1} \quad (27)$$

Check that x3a=x3:

```
> simplify(x3a-x3);
```

$$0 \quad (28)$$

Solve B2b=0 for z1 as function of y2:

```
> z1a:=solve(B2b,z1);
```

$$z1a := -\frac{1}{(R0+\alpha)^3\,(R0-1)}\left(y2\,R0^4-2\,y2\,R0^3+y2\,R0^2+2\,y2\,R0^3\,\alpha-4\,y2\,R0^2\,\alpha+2\,y2\,R0\,\alpha+y2\,R0^2\,\alpha^2-2\,y2\,R0\,\alpha^2+y2\,\alpha^2-R0^2\,\alpha^2-2\,R0^2\,\alpha-R0^2\right) \quad (29)$$

Simplify:

```
> z1:=(alpha+1)^2*R0^2/(R0+alpha)^3/(R0-1) - y2*(R0-1)/(R0+alpha);
```

$$z1 := \frac{(\alpha+1)^2\,R0^2}{(R0+\alpha)^3\,(R0-1)}-\frac{y2\,(R0-1)}{R0+\alpha} \quad (30)$$

Check that z1a=z1:

```
> simplify(z1a-z1);
```

$$0 \quad (31)$$

Solve Ca1 for y2:

```
> y2:=solve(Ca1,y2);
```

$$y2 := -\frac{(\alpha+1)\,R0\,(-4\,R0\,\alpha-4\,R0+R0^2+\alpha)}{(R0+\alpha)^2\,(R0-1)^2} \quad (32)$$

Use this y2-value in the expressions for x3 and z1.

```
> x3;
```

$$-\frac{(\alpha+1)^2\,R0^2}{(R0-1)^3\,(R0+\alpha)}+\frac{(\alpha+1)\,R0\,(-4\,R0\,\alpha-4\,R0+R0^2+\alpha)}{(R0+\alpha)\,(R0-1)^3} \quad (33)$$

```
> x3:=simplify(x3);
```

$$x3 := \frac{(\alpha+1)\,R0\,(-5\,R0\,\alpha-5\,R0+R0^2+\alpha)}{(R0-1)^3\,(R0+\alpha)} \quad (34)$$

```
> z1;
```

(35)

$$\frac{(\alpha+1)^2 R0^2}{(R0+\alpha)^3 (R0-1)} + \frac{(\alpha+1) R0 (-4 R0 \alpha - 4 R0 + R0^2 + \alpha)}{(R0+\alpha)^3 (R0-1)} \quad (35)$$

```
> z1:=simplify(z1);
```

$$z1 := \frac{(\alpha+1) R0 (-3 R0 \alpha - 3 R0 + R0^2 + \alpha)}{(R0+\alpha)^3 (R0-1)} \quad (36)$$

Summarize result:

```
> x1; x2; x3; y1; y2; z1;
```

$$\frac{R0-1}{R0+\alpha}$$

$$-\frac{R0 (\alpha+1)}{(R0+\alpha) (R0-1)}$$

$$\frac{(\alpha+1) R0 (-5 R0 \alpha - 5 R0 + R0^2 + \alpha)}{(R0-1)^3 (R0+\alpha)}$$

$$\frac{R0 (\alpha+1)}{(R0+\alpha)^2}$$

$$-\frac{(\alpha+1) R0 (-4 R0 \alpha - 4 R0 + R0^2 + \alpha)}{(R0+\alpha)^2 (R0-1)^2}$$

$$\frac{(\alpha+1) R0 (-3 R0 \alpha - 3 R0 + R0^2 + \alpha)}{(R0+\alpha)^3 (R0-1)} \quad (37)$$

Bartlett's version 2, based on Renshaw (3.5.38)

```
> restart;
```

The first two cumulant equations used by Renshaw are found by putting the derivatives in (3.4.13), (3.4.14) equal to zero. The third one is formulated after (3.5.38).
I denote them by A, B, Cb:

```
> A:=(a-b*kappa1)*kappa1 - b*kappa2;
  B:=(c-d*kappa1)*kappa1 + (2*a-d-4*b*kappa1)*kappa2 - 2*b*kappa3;
  Cb:=a*(kappa3+kappa1*kappa2) - b*(kappa4+3*kappa2^2+2*kappa1*
  kappa3+kappa1^2*kappa2) + c*kappa2
      - d*(kappa3+2*kappa1*kappa2);
```

$$A := (a - b \kappa1) \kappa1 - b \kappa2$$

$$B := (c - d \kappa1) \kappa1 + (2 a - d - 4 b \kappa1) \kappa2 - 2 b \kappa3$$

$$Cb := a (\kappa3 + \kappa1 \kappa2) - b (\kappa4 + 3 \kappa2^2 + 2 \kappa1 \kappa3 + \kappa1^2 \kappa2) + c \kappa2 - d (\kappa3 + 2 \kappa1 \kappa2) \quad (38)$$

```
> A:=map(simplify,collect(A,[a,b]));
  B:=map(simplify,collect(B,[a,b,c,d]));
  Cb:=map(simplify,collect(Cb,[a,b,c,d]));
```

$$A := \kappa1 a - (\kappa1^2 + \kappa2) b$$

$$B := 2\,\kappa2\,a - 2\,(\kappa3 + 2\,\kappa1\,\kappa2)\,b + \kappa1\,c - \left(\kappa1^2 + \kappa2\right)\,d$$

$$Cb := a\,(\kappa3 + \kappa1\,\kappa2) - b\left(\kappa4 + 3\,\kappa2^2 + 2\,\kappa1\,\kappa3 + \kappa1^2\,\kappa2\right) + c\,\kappa2 - d\,(\kappa3 + 2\,\kappa1\,\kappa2) \qquad \textbf{(39)}$$

Reparametrize with s=1 and mu=1:

```
> a:=R0-1;
  b:=(R0+alpha)/N;
  c:=R0+1;
  d:=(R0-alpha)/N;
```

$$a := R0 - 1$$

$$b := \frac{R0 + \alpha}{N}$$

$$c := R0 + 1$$

$$d := \frac{R0 - \alpha}{N} \qquad \textbf{(40)}$$

```
> A;
  B;
  Cb;
```

$$\kappa1\,(R0 - 1) - \frac{\left(\kappa1^2 + \kappa2\right)(R0 + \alpha)}{N}$$

$$2\,\kappa2\,(R0 - 1) - \frac{2\,(\kappa3 + 2\,\kappa1\,\kappa2)\,(R0 + \alpha)}{N} + \kappa1\,(R0 + 1) - \frac{\left(\kappa1^2 + \kappa2\right)(R0 - \alpha)}{N}$$

$$(R0 - 1)\,(\kappa3 + \kappa1\,\kappa2) - \frac{(R0 + \alpha)\left(\kappa4 + 3\,\kappa2^2 + 2\,\kappa1\,\kappa3 + \kappa1^2\,\kappa2\right)}{N} + (R0 + 1)\,\kappa2 - \frac{(R0 - \alpha)\,(\kappa3 + 2\,\kappa1\,\kappa2)}{N} \qquad \textbf{(41)}$$

Ansats:

```
> kappa1:=x1*N+x2+x3/N;
  kappa2:=y1*N+y2;
  kappa3:=z1*N;
  kappa4:=u1*N;
```

$$\kappa1 := x1\,N + x2 + \frac{x3}{N}$$

$$\kappa2 := y1\,N + y2$$

$$\kappa3 := z1\,N$$

$$\kappa4 := u1\,N \qquad \textbf{(42)}$$

```
> A; B; Cb;
```

$$\left(x1\,N + x2 + \frac{x3}{N}\right)(R0 - 1) - \frac{\left(\left(x1\,N + x2 + \frac{x3}{N}\right)^2 + y1\,N + y2\right)(R0 + \alpha)}{N}$$

$$2\,(y1\,N+y2)\,(R0-1)-\frac{2\left(z1\,N+2\left(x1\,N+x2+\frac{x3}{N}\right)(y1\,N+y2)\right)(R0+\alpha)}{N}$$
$$+\left(x1\,N+x2+\frac{x3}{N}\right)(R0+1)-\frac{\left(\left(x1\,N+x2+\frac{x3}{N}\right)^2+y1\,N+y2\right)(R0-\alpha)}{N}$$
$$(R0-1)\left(z1\,N+\left(x1\,N+x2+\frac{x3}{N}\right)(y1\,N+y2)\right)-\frac{1}{N}\Bigg((R0+\alpha)\bigg(u1\,N+3\,(y1\,N$$
$$+y2)^2+2\left(x1\,N+x2+\frac{x3}{N}\right)z1\,N+\left(x1\,N+x2+\frac{x3}{N}\right)^2(y1\,N+y2)\bigg)\Bigg)+(R0$$
$$+1)\,(y1\,N+y2)-\frac{(R0-\alpha)\left(z1\,N+2\left(x1\,N+x2+\frac{x3}{N}\right)(y1\,N+y2)\right)}{N} \qquad \textbf{(43)}$$

```
> A1:=coeff(A,N,1);
  A2:=coeff(A,N,0);
  A3:=coeff(A,N,-1);
```

$$A1 := x1\,(R0-1)-x1^2\,(R0+\alpha)$$
$$A2 := x2\,(R0-1)-(2\,x2\,x1+y1)\,(R0+\alpha)$$
$$A3 := x3\,(R0-1)-(2\,x3\,x1+x2^2+y2)\,(R0+\alpha) \qquad \textbf{(44)}$$

```
> B1:=coeff(B,N,1);
  B2:=coeff(B,N,0);
```

$$B1 := 2\,y1\,(R0-1)-4\,x1\,y1\,(R0+\alpha)+x1\,(R0+1)-x1^2\,(R0-\alpha)$$
$$B2 := 2\,y2\,(R0-1)-2\,(z1+2\,x2\,y1+2\,x1\,y2)\,(R0+\alpha)+x2\,(R0+1)-(2\,x2\,x1$$
$$+y1)\,(R0-\alpha) \qquad \textbf{(45)}$$

```
> Cb2:=coeff(Cb,N,2);
  Cb1:=coeff(Cb,N,1);
```

$$Cb2 := (R0-1)\,x1\,y1-(R0+\alpha)\,x1^2\,y1$$
$$Cb1 := (R0-1)\,(z1+x2\,y1+x1\,y2)-(R0+\alpha)\,(3\,y1^2+2\,x1\,z1+2\,x2\,x1\,y1$$
$$+x1^2\,y2)+(R0+1)\,y1-2\,(R0-\alpha)\,x1\,y1 \qquad \textbf{(46)}$$

Solve A1 for x1, B1 for y1, A2 for x2:

```
> x1a:=solve(A1,x1);
  x1:=x1a[2];
```

$$x1a := 0,\ \frac{R0-1}{R0+\alpha}$$
$$x1 := \frac{R0-1}{R0+\alpha} \qquad \textbf{(47)}$$

```
> y1:=solve(B1,y1);
```

(48)

$$y1 := \frac{R0\,(\alpha+1)}{(R0+\alpha)^2} \quad (48)$$

```
> x2:=solve(A2,x2);
```

$$x2 := -\frac{R0\,(\alpha+1)}{(R0+\alpha)\,(R0-1)} \quad (49)$$

Show Cb2=0:

```
> Cb2;
```

$$0 \quad (50)$$

Three equations remain to solve: A3=0, B2=0, Cb1=0. I simplify A3, B2, Cb1:

```
> A3;
```

$$x3\,(R0-1) - \left(\frac{2\,x3\,(R0-1)}{R0+\alpha} + \frac{R0^2\,(\alpha+1)^2}{(R0+\alpha)^2\,(R0-1)^2} + y2\right)(R0+\alpha) \quad (51)$$

Simplify:

```
> A3a:=collect(A3,[x3,y2]);
```

$$A3a := (-R0+1)\,x3 + (-R0-\alpha)\,y2 - \frac{R0^2\,(\alpha+1)^2}{(R0+\alpha)\,(R0-1)^2} \quad (52)$$

Show that A3a=A3:

```
> simplify(A3-A3a);
```

$$0 \quad (53)$$

```
> B2;
```

$$2\,y2\,(R0-1) - 2\left(z1 - \frac{2\,R0^2\,(\alpha+1)^2}{(R0+\alpha)^3\,(R0-1)} + \frac{2\,(R0-1)\,y2}{R0+\alpha}\right)(R0+\alpha) - \frac{R0\,(\alpha+1)\,(R0+1)}{(R0+\alpha)\,(R0-1)} + \frac{R0\,(\alpha+1)\,(R0-\alpha)}{(R0+\alpha)^2} \quad (54)$$

Simplify:

```
> B2a:=collect(B2,[y2,z1]);
```

$$B2a := (-2\,R0+2)\,y2 + (-2\,R0-2\,\alpha)\,z1 + \frac{R0\,(\alpha+1)\,(R0-\alpha)}{(R0+\alpha)^2} + \frac{4\,R0^2\,(\alpha+1)^2}{(R0+\alpha)^2\,(R0-1)} - \frac{R0\,(\alpha+1)\,(R0+1)}{(R0+\alpha)\,(R0-1)} \quad (55)$$

```
> B2b:=op(1,B2a)+op(2,B2a)+simplify(op(3,B2a)+op(4,B2a)+op(5,B2a));
```

$$B2b := (-2\,R0+2)\,y2 + (-2\,R0-2\,\alpha)\,z1 + \frac{2\,R0^2\,(\alpha+1)^2}{(R0+\alpha)^2\,(R0-1)} \quad (56)$$

Show that B2=B2b:

```
> simplify(B2-B2b);
```

$$0 \quad (57)$$

```
> Cb1;
```

$$(R0-1)\left(z1-\frac{R0^2(\alpha+1)^2}{(R0+\alpha)^3(R0-1)}+\frac{(R0-1)\,y2}{R0+\alpha}\right)-(R0+\alpha)\left(\frac{R0^2(\alpha+1)^2}{(R0+\alpha)^4}+\frac{2\,(R0-1)\,z1}{R0+\alpha}+\frac{(R0-1)^2\,y2}{(R0+\alpha)^2}\right)+\frac{(R0+1)\,R0\,(\alpha+1)}{(R0+\alpha)^2}-\frac{2\,(R0-\alpha)\,(R0-1)\,R0\,(\alpha+1)}{(R0+\alpha)^3} \quad (58)$$

Simplify:

```
> Cb1a:=collect(Cb1,[y2,z1]);
```

$$Cb1a:=(-R0+1)\,z1-\frac{2\,R0^2(\alpha+1)^2}{(R0+\alpha)^3}+\frac{(R0+1)\,R0\,(\alpha+1)}{(R0+\alpha)^2}-\frac{2\,(R0-\alpha)\,(R0-1)\,R0\,(\alpha+1)}{(R0+\alpha)^3} \quad (59)$$

```
> Cb1b:=op(1,Cb1a) + simplify(Cb1a-op(1,Cb1a));
```

$$Cb1b:=(-R0+1)\,z1-\frac{R0\,(\alpha+1)\,(R0^2-R0-\alpha\,R0+\alpha)}{(R0+\alpha)^3} \quad (60)$$

Show Cb1=Cb1b:

```
> simplify(Cb1-Cb1b);
```

$$0 \quad (61)$$

Solve Cb1b for z1:

```
> z1:=solve(Cb1b,z1);
```

$$z1:=-\frac{(R0-\alpha)\,R0\,(\alpha+1)}{(R0+\alpha)^3} \quad (62)$$

```
> y2:=solve(B2b,y2);
```

$$y2:=\frac{(R0^2+\alpha)\,R0\,(\alpha+1)}{(R0+\alpha)^2\,(R0-1)^2} \quad (63)$$

```
> x3:=solve(A3a,x3);
```

$$x3:=-\frac{(R0+1)\,R0\,(\alpha+1)}{(R0-1)^3} \quad (64)$$

Summarize:

```
> x1; x2; x3; y1; y2; z1;
```

$$\frac{R0-1}{R0+\alpha}$$

$$-\frac{R0\,(\alpha+1)}{(R0+\alpha)\,(R0-1)}$$

$$-\frac{(R0+1)\,R0\,(\alpha+1)}{(R0-1)^3}$$

$$\frac{R0\,(\alpha+1)}{(R0+\alpha)^2}$$

$$\frac{(R0^2+\alpha)\,R0\,(\alpha+1)}{(R0+\alpha)^2\,(R0-1)^2}$$

$$-\frac{(R0-\alpha)\,R0\,(\alpha+1)}{(R0+\alpha)^3} \quad \textbf{(65)}$$

Analysis of the results in Bhowmick et al. (2016), Section 4, with s=1.

```
> restart;
```

The function H is defined by

```
> H:=(alpha+1)*R0/(R0-1)^2;
```

$$H := \frac{(\alpha+1)\,R0}{(R0-1)^2} \quad \textbf{(66)}$$

The one-term asymptotic approximation of the mean is denoted by m1:

```
> m1:=(R0-1)*N/(R0+alpha);
```

$$m1 := \frac{(R0-1)\,N}{R0+\alpha} \quad \textbf{(67)}$$

The mean m and the variance V are given in (24) and (25) as follows:

```
> m:=m1/(1+H/N);
  V:=m^2*H/N;
```

$$m := \frac{(R0-1)\,N}{(R0+\alpha)\left(1+\frac{(\alpha+1)\,R0}{(R0-1)^2\,N}\right)}$$

$$V := \frac{N\,(\alpha+1)\,R0}{(R0+\alpha)^2\left(1+\frac{(\alpha+1)\,R0}{(R0-1)^2\,N}\right)^2} \quad \textbf{(68)}$$

I derive asymptotic approximations of m and V. I include three terms in the asymptotic approximation of m, and two terms in the asymptotic approxiamtion of V:

```
> ma:=convert(asympt(m,N,2),polynom);
```

$$ma := \frac{(R0-1)\,N}{R0+\alpha} - \frac{(\alpha+1)\,R0}{(R0-1)\,(R0+\alpha)} + \frac{(\alpha+1)^2\,R0^2}{(R0-1)^3\,(R0+\alpha)\,N} \quad \textbf{(69)}$$

```
> Va:=convert(asympt(V,N,2),polynom);
```

$$Va := \frac{(\alpha+1)\ R0\ N}{(R0+\alpha)^2} - \frac{2\ (\alpha+1)^2\ R0^2}{(R0+\alpha)^2\ (R0-1)^2} \quad \textbf{(70)}$$

To find an expression for the third cumulant, I introduce the four quantities A, B, C, DD:

```
> A:=a-b*m;
  B:=c/m-2*d;
  C:=a/m - 2*b;
  DD:=-d/m;
```

$$A := a - \frac{b\ (R0-1)\ N}{(R0+\alpha)\left(1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}\right)}$$

$$B := \frac{c\ (R0+\alpha)\left(1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}\right)}{(R0-1)\ N} - 2\ d$$

$$C := \frac{a\ (R0+\alpha)\left(1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}\right)}{(R0-1)\ N} - 2\ b$$

$$DD := -\frac{d\ (R0+\alpha)\left(1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}\right)}{(R0-1)\ N} \quad \textbf{(71)}$$

After reparametrization I get:

```
> a:=R0-1; b:=(R0+alpha)/N; c:=R0+1; d:=(R0-alpha)/N;
```

$$a := R0-1$$

$$b := \frac{R0+\alpha}{N}$$

$$c := R0+1$$

$$d := \frac{R0-\alpha}{N} \quad \textbf{(72)}$$

```
> A; B; C; DD;
```

$$R0-1-\frac{R0-1}{1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}}$$

$$\frac{(R0+1)\ (R0+\alpha)\left(1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}\right)}{(R0-1)\ N} - \frac{2\ (R0-\alpha)}{N}$$

$$\frac{(R0+\alpha)\left(1+\frac{(\alpha+1)\ R0}{(R0-1)^2\ N}\right)}{N} - \frac{2\ (R0+\alpha)}{N}$$

$$-\frac{(R0-\alpha)\,(R0+\alpha)\left(1+\frac{(\alpha+1)\,R0}{(R0-1)^2\,N}\right)}{N^2\,(R0-1)} \quad (73)$$

One-term asymptotic approximations of these quantities are:

```
> Aa:=simplify(convert(asympt(A,N,2),polynom));
```

$$Aa := \frac{(\alpha+1)\,R0}{N\,(R0-1)} \quad (74)$$

```
> Ba:=simplify(convert(asympt(B,N,2),polynom));
```

$$Ba := -\frac{R0^2-3\,R0\,\alpha-3\,R0+\alpha}{N\,(R0-1)} \quad (75)$$

```
> Ca:=simplify(convert(asympt(C,N,2),polynom));
```

$$Ca := -\frac{R0+\alpha}{N} \quad (76)$$

```
> DDa:=simplify(convert(asympt(DD,N,3),polynom));
```

$$DDa := -\frac{R0^2-\alpha^2}{(R0-1)\,N^2} \quad (77)$$

```
> simplify((Aa+Ba)/Ca);
```

$$\frac{-4\,R0\,\alpha-4\,R0+R0^2+\alpha}{(R0+\alpha)\,(R0-1)} \quad (78)$$

```
> mu3:=-(Aa+Ba)/(Ca+DDa)*Va;
```

$$\mu 3 := -\frac{1}{-\frac{R0+\alpha}{N}-\frac{R0^2-\alpha^2}{(R0-1)\,N^2}}\left(\left(\frac{(\alpha+1)\,R0}{N\,(R0-1)} - \frac{R0^2-3\,R0\,\alpha-3\,R0+\alpha}{N\,(R0-1)}\right)\left(\frac{(\alpha+1)\,R0\,N}{(R0+\alpha)^2}-\frac{2\,(\alpha+1)^2\,R0^2}{(R0+\alpha)^2\,(R0-1)^2}\right)\right) \quad (79)$$

```
> mu3a:=simplify(convert(asympt(mu3,N,2),polynom));
```

$$mu3a := -\frac{N\,R0\,(\alpha+1)\,\left(-4\,R0\,\alpha-4\,R0+R0^2+\alpha\right)}{(R0+\alpha)^3\,(R0-1)} \quad (80)$$